```
> restart: with(plots):
```

Possible functions that compactify infinite interval:

```
> F := [erf(x), tanh(x), x/sqrt(1+x^2), arctan(Pi/2*x)/(Pi/2)]; plot
  (F, x=-5..5);
```

$$F := \left[\operatorname{erf}(x), \tanh(x), \frac{x}{\sqrt{x^2+1}}, \frac{2\arctan\left(\frac{\pi x}{2}\right)}{\pi}\right]$$



The main difference between them is asymptotic behavior at infinity:

```
> for f in F do; f = series(f, x=infinity); end do;
```

$$\operatorname{erf}(x) = 1 + \frac{-\frac{1}{\sqrt{\pi}\,x} + \frac{1}{2\sqrt{\pi}\,x^3} - \frac{3}{4\sqrt{\pi}\,x^5} + O\left(\frac{1}{x^7}\right)}{e^{x^2}}$$

$$\tanh(x) = 1 - \frac{2}{(e^x)^2} + \frac{2}{(e^x)^4} - \frac{2}{(e^x)^6} + \frac{2}{(e^x)^8} - \frac{2}{(e^x)^{10}} + O\left(\frac{1}{(e^x)^{12}}\right)$$

$$\frac{x}{\sqrt{x^2+1}} = 1 - \frac{1}{2x^2} + \frac{3}{8x^4} + O\left(\frac{1}{x^6}\right)$$

$$\frac{2\arctan\left(\frac{\pi x}{2}\right)}{\pi} = 1 - \frac{4}{\pi^2 x} + \frac{16}{3\pi^4 x^3} - \frac{64}{5\pi^6 x^5} + O\left(\frac{1}{x^7}\right) \tag{1}$$

Let's pick one, say:

```
> Y := F[3]; X := simplify(solve(Y=cos(theta), x),symbolic);
```

$$Y := \frac{x}{\sqrt{x^2+1}}$$

$$X := \frac{\cos(\theta)}{\sin(\theta)} \tag{2}$$

Odd Chebyshev polynomials on compactified coordinate y look like this:

```
> plot([ChebyshevT(2*k+1, Y) $k=0..5], x=-10..10);
```



The basis functions are:

```
> T[n] := cos(n*arccos(x/sqrt(1^2+x^2)));
```

$$T_n := \cos\left(n \arccos\left(\frac{x}{\sqrt{x^2+1}}\right)\right) \quad \textbf{(3)}$$

They are best written in terms of trigonometric argument theta:

```
> 'T[n]' = simplify(subs(x=X, %), symbolic);
> diff(T[n],x): T[n,x] = simplify(subs(x=X, %), symbolic);
> diff(T[n],x$2): simplify(%): simplify(subs(x=X, %),symbolic): T[n,
  xx] = simplify(%,trig);
```

$$T_n = \cos(n\,\theta)$$

$$T_{n,\,x} = n \sin(n\,\theta) \sin(\theta)^2$$

$$T_{n,\,xx} = -\frac{n\left(\sin(\theta)\cos(n\,\theta)\,n + 2\cos(\theta)\sin(n\,\theta)\right)\sin(\theta)^4}{\sqrt{\sin(\theta)^2}} \quad \textbf{(4)}$$